

NAG Library Routine Document

G05TDF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G05TDF generates a vector of pseudorandom integers from a discrete distribution with a given PDF (probability density function) or CDF (cumulative distribution function) p .

2 Specification

```
SUBROUTINE G05TDF(MODE, N, P, NP, IP1, ITYPE, R, LR, STATE, X, IFAIL)
INTEGER          MODE, N, NP, IP1, ITYPE, LR, STATE(lstate), X(N), IFAIL
double precision P(NP), R(LR)
```

3 Description

G05TDF generates a sequence of n integers x_i , from a discrete distribution defined by information supplied in P. This may either be the PDF or CDF of the distribution. A reference vector is first set up to contain the CDF of the distribution in its higher elements, followed by an index.

Setting up the reference vector and subsequent generation of variates can each be performed by separate calls to G05TDF or may be combined in a single call.

One of the initialization routines G05KFF (for a repeatable sequence if computed sequentially) or G05KGF (for a non-repeatable sequence) must be called prior to the first call to G05TDF.

4 References

Kendall M G and Stuart A (1969) *The Advanced Theory of Statistics (Volume 1)* (3rd Edition) Griffin
 Knuth D E (1981) *The Art of Computer Programming (Volume 2)* (2nd Edition) Addison–Wesley

5 Parameters

- 1: MODE – INTEGER *Input*
On entry: a code for selecting the operation to be performed by the routine.
 MODE = 0
 Set up reference vector only.
 MODE = 1
 Generate variates using reference vector set up in a prior call to G05TDF.
 MODE = 2
 Set up reference vector and generate variates.
Constraint: MODE = 0, 1 or 2.
- 2: N – INTEGER *Input*
On entry: n , the number of pseudorandom numbers to be generated.
Constraint: $N \geq 0$.

- 3: $P(NP)$ – **double precision** array *Input*
On entry: the PDF or CDF of the distribution.
Constraints:
 $0 \leq P(i) \leq 1$, for $i = 1, 2, \dots, NP$;
 if $ITYPE = 1$, $\sum_{i=1}^{NP} P(i) = 1.0$;
 if $ITYPE = 2$, $P(i) < P(j)$, $i < j$ and $P(NP) = 1.0$.
- 4: NP – INTEGER *Input*
On entry: the number of values supplied in P defining the PDF or CDF of the discrete distribution.
Constraint: $NP > 0$.
- 5: $IP1$ – INTEGER *Input*
On entry: the value of the variate, a whole number, to which the probability in $P(1)$ corresponds.
- 6: $ITYPE$ – INTEGER *Input*
On entry: indicates the type of information contained in P .
 $ITYPE = 1$
 P contains a probability distribution function (PDF).
 $ITYPE = 2$
 P contains a cumulative distribution function (CDF).
Constraint: $ITYPE = 1$ or 2 .
- 7: $R(LR)$ – **double precision** array *Communication Array*
On entry: if $MODE = 1$, the reference vector from the previous call to G05TDF.
On exit: the reference vector.
- 8: LR – INTEGER *Input*
On entry: the dimension of the array R as declared in the (sub)program from which G05TDF is called.
Suggested value: $LR = 10 + 1.4 \times NP$ approximately (for optimum efficiency in generating variates).
Constraint: $LR \geq NP + 8$.
- 9: $STATE(lstate)$ – INTEGER array *Communication Array*
Note: $lstate = LSTATE$, the dimension of $STATE$ as supplied to the initialization routine G05KFF or G05KGF.
On entry: contains information on the selected base generator and its current state.
On exit: contains updated information on the state of the generator.
- 10: $X(N)$ – INTEGER array *Output*
On exit: contains n pseudorandom numbers from the specified discrete distribution.
- 11: $IFAIL$ – INTEGER *Input/Output*
On entry: $IFAIL$ must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.
On exit: $IFAIL = 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0 . **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry $IFAIL = 0$ or -1 , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

$IFAIL = 1$

On entry, $MODE \neq 0, 1$ or 2 .

$IFAIL = 2$

On entry, $N < 0$.

$IFAIL = 3$

With $ITYPE = 1$, $P(j) < 0$ for at least one value of j .

With $ITYPE = 1$, the sum of $P(j)$, for $j = 1, 2, \dots, NP$, does not equal 1 .

With $ITYPE = 2$, the values of $P(j)$ are not all in non-descending order.

$IFAIL = 4$

On entry, $NP < 1$.

$IFAIL = 6$

On entry, $ITYPE \neq 1$ or 2 .

$IFAIL = 7$

The value of NP , $ITYPE$ or $IP1$ is not the same as when R was set up in a previous call to G05TDF with $MODE = 0$ or 2 .

On entry, the R vector was not initialized correctly, or has been corrupted.

$IFAIL = 8$

On entry, LR is too small when $MODE = 0$ or 2 .

$IFAIL = 9$

On entry, $STATE$ vector was not initialized or has been corrupted.

7 Accuracy

Not applicable.

8 Further Comments

None.

9 Example

This example prints 20 pseudorandom variates from a discrete distribution whose PDF, p , is defined as follows:

n	p
-5	0.01
-4	0.02
-3	0.04
-2	0.08
-1	0.20
0	0.30
1	0.20
2	0.08
3	0.04
4	0.02
5	0.01

The reference vector is set up and the variates are generated by a single call to G05TDF, after initialization by G05KFF.

9.1 Program Text

```

*      G05TDF Example Program Text
*      Mark 22 Release. NAG Copyright 2006.
*      .. Parameters ..
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
      INTEGER          NP, MSEED, MSTATE, N, LR
      PARAMETER        (NP=11,MSEED=1,MSTATE=633,N=20,LR=60)
*      .. Local Scalars ..
      INTEGER          GENID, I, IFAIL, IP1, ITYPE, LSEED, LSTATE, SUBID
*      .. Local Arrays ..
      DOUBLE PRECISION P(NP), R(LR)
      INTEGER          SEED(MSEED), STATE(MSTATE), X(N)
*      .. External Subroutines ..
      EXTERNAL         G05KFF, G05TDF
*      .. Data statements ..
      DATA             P/0.01D0, 0.02D0, 0.04D0, 0.08D0, 0.2D0, 0.3D0,
+                      0.2D0, 0.08D0, 0.04D0, 0.02D0, 0.01D0/
*      .. Executable Statements ..
      WRITE (NOUT,*) 'G05TDF Example Program Results'
      WRITE (NOUT,*)
*      Set the lowest value for the distribution
      IP1 = -5
*      P is a PDF
      ITYPE = 1
*      Initialize the seed
      SEED(1) = 1762543
*      GENID and SUBID identify the base generator
      GENID = 1
      SUBID = 1
*      Initialize the generator to a repeatable sequence
      LSTATE = MSTATE
      LSEED = MSEED
      IFAIL = 1
      CALL G05KFF(GENID,SUBID,SEED,LSEED,STATE,LSTATE,IFAIL)
      IF (IFAIL.NE.0) THEN
         WRITE (NOUT,99998) IFAIL
         GO TO 20
      END IF
*      Generate the variates
*      Choose MODE = 2
      IFAIL = 1
      CALL G05TDF(2,N,P,NP,IP1,ITYPE,R,LR,STATE,X,IFAIL)
      IF (IFAIL.NE.0) THEN
         WRITE (NOUT,99997) IFAIL
         GO TO 20
      END IF
*      Display the variates

```

```
      WRITE (NOUT,99999) (X(I),I=1,N)
*
      20 CONTINUE
*
99999 FORMAT (1X,I12)
99998 FORMAT (1X,' ** G05KFF returned with IFAIL = ',I5)
99997 FORMAT (1X,' ** G05TDF returned with IFAIL = ',I5)
      END
```

9.2 Program Data

None.

9.3 Program Results

G05TDF Example Program Results

```
0
-2
1
1
-2
0
0
1
0
1
-3
-1
0
-3
0
-1
-1
5
2
0
```
